

```

In[100]:= $Assumptions = e > 0 && r > 0 && μ > 0 && ν ≥ 0 && ν ≤ 0.5 &&
          t > 0 && r ∈ Reals && e ∈ Reals && μ ∈ Reals && ν ∈ Reals && t ∈ Reals;

In[101]:= re[r_] := Sqrt[r^2 + e^2];

In[102]:= β := Sqrt[μ];

In[103]:= α := β * Sqrt[1 + 1/(1 - 2*ν)];

In[104]:= (*Pseudo-Potentials*)

In[105]:= Q[r_, s_, w_] :=
          2*r^3/re[r] - s*w*(s/re[s] + w/re[w]) + e^2*(s - r)*(1/re[s] - 1/re[w]);

In[106]:= U[r_, t_, c_] := Q[r, r + c*t, r - c*t] / (16*Pi*c^2*r^3);

In[107]:= dQ[r_, s_, w_] := 2*r^2*(3/re[r] - r^2/re[r]^3) -
          2*r*(s/re[s] + w/re[w]) - e^2*r*(s/re[s]^3 + w/re[w]^3);

In[108]:= dU[r_, t_, c_] :=
          (dQ[r, r + c*t, r - c*t] - (3/r)*Q[r, r + c*t, r - c*t]) / (16*Pi*c^2*r^3);

In[109]:= (*Push Dynamic Kelvinlet*)

In[110]:= A[r_, t_] := U[r, t, α] + 2*U[r, t, β] + r*dU[r, t, β];

In[111]:= B[r_, t_] := (dU[r, t, α] - dU[r, t, β])/r;

In[112]:= (*Regularized Kelvinlet*)
          a := 1/(4*Pi*μ);

In[113]:= b := a/(4*(1 - ν));

In[114]:= AA[r_] := (a - b)/re[r] + (a/2)*(e^2/re[r]^3);

In[115]:= BB[r_] := b/re[r]^3;

In[116]:= (*Verification*)

In[117]:= FullSimplify[Limit[A[r, t], t → Infinity] - AA[r]]
Out[117]= 0

In[118]:= FullSimplify[Limit[B[r, t], t → Infinity] - BB[r]]
Out[118]= 0

```